TMA4195: 2004-09-08

Toy problem 2 revisited

```
> restart:
> epsilon*diff(y(x),x,x)+diff(y(x),x)+y(x)=0,y(0)=0,y(1)=1; toy2:=%:
```

$$\varepsilon \left(\frac{d^2}{dx^2} y(x)\right) + \left(\frac{d}{dx} y(x)\right) + y(x) = 0,\ y(0) = 0,\ y(1) = 1$$

```
> dsolve([toy2],y(x));
  exact2:=%:
```

$$y(x) = \frac{e^{\left(\frac{(-1+\sqrt{1-4\,\varepsilon})\,x}{2\,\varepsilon}\right)}}{e^{\left(\frac{-1+\sqrt{1-4\,\varepsilon}}{2\,\varepsilon}\right)} - e^{\left(-\frac{1+\sqrt{1-4\,\varepsilon}}{2\,\varepsilon}\right)}} - \frac{e^{\left(-\frac{(1+\sqrt{1-4\,\varepsilon})\,x}{2\,\varepsilon}\right)}}{e^{\left(\frac{-1+\sqrt{1-4\,\varepsilon}}{2\,\varepsilon}\right)} - e^{\left(-\frac{1+\sqrt{1-4\,\varepsilon}}{2\,\varepsilon}\right)}}$$

```
> toy2[1],toy2[3];
  dsolve(subs(epsilon=0,[%]),y(x)): simplify(%);
  y0:=unapply(rhs(%),x):
```

$$\varepsilon \left(\frac{d^2}{dx^2} y(x)\right) + \left(\frac{d}{dx} y(x)\right) + y(x) = 0,\ y(1) = 1$$

$$y(x) = e^{(1-x)}$$

```
> diff(y1(x),x)+y1(x)=-y0(x);
  dsolve([%,y1(1)=0],y1(x));
  toy2y:=unapply(exp(1-x)+epsilon*rhs(%),x): y(x)=toy2y(x);
```

$$\left(\frac{d}{dx} y1(x)\right) + y1(x) = -e^{(1-x)}$$

$$y1(x) = (-e\,x + e)\,e^{(-x)}$$

$$y(x) = e^{(1-x)} + \varepsilon\,(-e\,x + e)\,e^{(-x)}$$

```
> Y0:=xi->exp(1)-exp(1-xi): Y(xi)=Y0(xi);
```

$$Y(\xi) = e - e^{(1-\xi)}$$

```
> diff(Y1(xi),xi,xi)+diff(Y1(xi),xi)=-Y0(xi),Y1(0)=0;
  dsolve([%],Y1(xi)): simplify(%);
  toy2Y:=unapply(exp(1)-exp(1-xi)+epsilon*rhs(%),xi):
  Y(xi)=toy2Y(xi);
```

$$\left(\frac{d^2}{d\xi^2} Y1(\xi)\right) + \left(\frac{d}{d\xi} Y1(\xi)\right) = -e + e^{(1-\xi)},\ Y1(0) = 0$$

$$Y1(\xi) = -\mathbf{e}\,\xi - \xi\,\mathbf{e}^{(1-\xi)} - \mathbf{e}^{(-\xi)}\,\_C2 + \_C2$$

$$Y(\xi) = \mathbf{e} - \mathbf{e}^{(1-\xi)} + \varepsilon\,(-\mathbf{e}\,\xi - \xi\,\mathbf{e}^{(1-\xi)} - \mathbf{e}^{(-\xi)}\,\_C2 + \_C2)$$

```
> taylor(toy2y(x),x,2);
  toy2Y(Psi*X/epsilon)=subs(x=Psi*X,%);
```

$$(\mathbf{e} + \varepsilon\,\mathbf{e}) + (-\mathbf{e} - 2\,\varepsilon\,\mathbf{e})\,x + O(x^2)$$

$$\mathbf{e} - \mathbf{e}^{\left(1 - \frac{\Psi X}{\varepsilon}\right)} + \varepsilon\left(-\frac{\mathbf{e}\,\Psi X}{\varepsilon} - \frac{\Psi X\,\mathbf{e}^{\left(1 - \frac{\Psi X}{\varepsilon}\right)}}{\varepsilon} - \mathbf{e}^{\left(-\frac{\Psi X}{\varepsilon}\right)}\,\_C2 + \_C2\right) = \mathbf{e} + \varepsilon\,\mathbf{e} + (-\mathbf{e} - 2\,\varepsilon\,\mathbf{e})\,\Psi$$
$$+ O(\Psi^2 X^2)$$

```
> expand(%);
```

$$\mathbf{e} - \frac{\mathbf{e}}{\mathbf{e}^{\left(\frac{\Psi X}{\varepsilon}\right)}} - \mathbf{e}\,\Psi X - \frac{\Psi X\,\mathbf{e}}{\mathbf{e}^{\left(\frac{\Psi X}{\varepsilon}\right)}} - \frac{\varepsilon\,\_C2}{\mathbf{e}^{\left(\frac{\Psi X}{\varepsilon}\right)}} + \varepsilon\,\_C2 = \mathbf{e} + \varepsilon\,\mathbf{e} - \mathbf{e}\,\Psi X - 2\,\Psi X\,\varepsilon\,\mathbf{e} + O(\Psi^2 X^2)$$

```
> param:=epsilon=0.1,_C2=exp(1):
  plots[display](
    plot(subs(exact2,param,y(x)),x=0..1,color=red),

  plot(subs(xi=x/epsilon,param,y0(x)+Y0(xi)-exp(1)),x=0..1,color=blue),

  plot(subs(xi=x/epsilon,param,toy2y(x)+toy2Y(xi)-((1-x+epsilon)*exp(1)
  )),
      x=0..1,color=green));
```




```
>
```

Enzyme kinetics

```
> restart:
```

I could have let Maple do more of the calculation, but my only purpose here is to plot exact versus approximate solutions.

```
> param:=epsilon=eta/(sigma+1),sigma=1,eta=0.2,kappa=1:
  param, epsilon=subs(param,epsilon);
```

$$\varepsilon = \frac{\eta}{\sigma+1}, \sigma = 1, \eta = 0.2, \kappa = 1, \varepsilon = 0.1000000000$$

```
> 1/((kappa+1)*(sigma+1))*diff(s(t),t)
  =-s(t)+(sigma/(sigma+1)*s(t)+kappa/((kappa+1)*(sigma+1)))*c(t);
  epsilon/((kappa+1)*(sigma+1))*diff(c(t),t)
  =s(t)-(sigma*s(t)+1)/(sigma+1)*c(t);
  ode:=%%,%:
```

$$\frac{\frac{d}{dt}\, s(t)}{(\kappa+1)\,(\sigma+1)} = -s(t) + \left(\frac{\sigma\, s(t)}{\sigma+1} + \frac{\kappa}{(\kappa+1)\,(\sigma+1)}\right) c(t)$$

$$\frac{\varepsilon\left(\frac{d}{dt}c(t)\right)}{(\kappa+1)(\sigma+1)} = s(t) - \frac{(\sigma\, s(t)+1)\, c(t)}{\sigma+1}$$

```
> initial:=s(0)=1,c(0)=0: initial; depvars:=s(t),c(t):
```

$$s(0)=1,\, c(0)=0$$

```
> dsolve(subs(param,[ode,initial]),numeric,[depvars],output=listprocedure);
sols:=subs(%,s(t)): solc:=subs(%%,c(t)):
```

$$[t = \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc};,\, s(t) = \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc};,\, c(t) = \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc};]$$

```
> plots[display](
    plot(sols(t),t=0..2,color=black),
    plot(solc(t),t=0..2,color=blue));
  exactplot:=%:
```




```
> (sigma+1)*s/(sigma*s+1)-exp(-(kappa+1)*(sigma+1)/epsilon*t);
pertc:=unapply(subs(param,%),[s,t]);
```

$$\frac{(\sigma+1)\, s}{\sigma\, s+1} - \mathbf{e}^{\left(-\frac{(\kappa+1)(\sigma+1)\, t}{\varepsilon}\right)}$$

$$pertc := (s, t) \to \frac{2\,s}{s+1} - \mathbf{e}^{(-40.00000000\,t)}$$

```
> (1-sigma*s-ln(s))/(sigma+1);
  pertt:=unapply(subs(param,%),t);
```

$$\frac{1 - \sigma\,s - \ln(s)}{\sigma + 1}$$

$$pertt := t \to \frac{1}{2} - \frac{1}{2}\,s - \frac{1}{2}\ln(s)$$

```
> plots[display](
    plot([pertt(s),s,s=0.2..1],color=red),
    plot([pertt(s),pertc(s,pertt(s)),s=0.2..1],color=green));
  pertplot:=%:
```




```
> plots[display]([exactplot,pertplot]);
```



1
0.8
0.6
0.4
0.2
0
0
0.5
1
1.5
2
t

>